

## 1. 系统概述

半导体的制造包括从硅晶体生长到最后的封装一系列复杂的自动化操作。一般各个部件的封装阶段指的是围绕一个硅模具的两个或更多的材质层间的接合与粘贴。我们关心的是这些层之间的粘性物质的粘贴过程。

这台工业机床应用在芯片封装材料的涂抹上（芯片封装材料——Chip Encapsulation Material, CEM）。当硅模具粘贴到封装基座之后，CEM 液体涂抹过程可以在封装芯片之间制造一个粘结层。液体的种类包括陶瓷粘合剂，导电及非导电环氧物质，热接触面材料，以及焊膏。

液体涂抹工艺需要至少两个轴在一个坐标系内密切配合，第三个轴用来控制液体涂抹喷嘴的垂直位置，第四个轴带动传送带，第五个轴控制液泵的液体流量（图1）。复杂的基座表面往往需要喷嘴轴偶合成三维结构。极其复杂的表面可能需要多加两个轴在另一个坐标系中偶合，用来完成涂抹喷嘴轴的偏航控制。要达到精确的液体涂抹通常需要线性插补及电子齿轮运动模式。在这个应用实例中，我们将涉及到一个坐标系，4个运动控制轴外加一个控制流量的轴。尽管这是一个特例，但其中包含的信息亦可用在一般的液体涂抹应用中。




图 1.

## 2. 要求

下面给出了运动控制系统的最基本要求：

- 五个轴（XYZWE）
- 预定的 XY 坐标路径

- . 标粘合剂的量总保持一致（精确流量控制）
- . I/O
  - 新载物盘传感器（数字输入）
  - 载物盘到位传感器（数字输入）
  - 载物盘空载传感器（数字输入）
  - 液位监测（模拟输入）
  - 液位低报警（模拟输出）
  - XYZ 轴的正 / 反向限位（数字输入）
  - W 轴的零点开关（数字输入）
- (5). 标终端用户界面
  - 操作员输入基座的深度
  - 开始涂抹周期
  - 监测周期中机床状态

### 3. 部件的选择

此段描述了为实现晶片定位机控制系统，您可以选择的 GALIL 的硬件及软件产品。下面是一个主要部件的列表，后面附有简单描述。

表 1. 液体涂抹机控制系统需要的材料清单

部件名称	描述	价格 (U.S.) 单买 / 批量
<a href="#">DMC-1850</a>	PCI 总线 5 轴运动控制器	\$2595 / \$1345
<a href="#">CABLE-100-1m</a>	1米长高密度电缆	\$125 / \$95 x2
<a href="#">AMP-19540</a>	可驱动四台有刷或无刷伺服电机的放大器	\$795/ \$495
<a href="#">BLM-N23-50-1000</a> 或同等产品	带有 100 线编码器及 Hall 传感器的 Nema 23 无刷伺服电机	请咨询生产商
<a href="#">CPS-12-56</a> 或同等产品	12A , 56V 电源	请咨询生产商
<a href="#">ICM-2900-FL</a>	ICM-2900 带法兰	\$295 / \$195
<a href="#">-OPTO Option</a>	指明输出是否加光电隔离	+\$50
<a href="#">CB-50-100-1880</a>	100 针高密度线 / 带形线转接器	\$75 / \$50
<a href="#">WSDK Servo Tuning Software</a>	伺服调试及分析软件	\$195 (one time)

#### 控制器：DMC-1850

因为此应用中需要终端用户界面，所以需要一台专用的 PC 机，加之需要对 5 轴控制，这使得 DMC-1850 PCI 总线控制器成为了最优选择。此外，DMC-1850 还为传感器提供了光电隔离输入，并为液面监测提供了模拟输入。




图 2. DMC-1850

#### 电机：BLM-N23-50-1000

为了达到免维护运行，我们选择无刷电机。由于晶片定位机需要不到 0.3Nm 的连续转矩，所以我们推荐使用 Galil 的 NEMA 23 #BLM-N23-50-1000 无刷电机或者其他同等机。

电机上装有一台每转 1000 脉冲的增量编码器，积分后每转产生 4000 个脉冲。由于增量编码器为输入信号提供了通向放大器的交换通道，电机上不许要安装 Hall 传感器。

#### 放大器：AMP-19540

为了驱动 X, Y, Z 和 W 轴电机，我们选择了非常紧凑的 AMP-19540，它是台无刷四轴放大器（每个轴的驱动能力为 500W）。第五轴液泵需要有一个独立的驱动设备，因此它通过 ICM-2900 和 CB-50-100 与控制器相连。它们用来扩展 5-8 轴控制器上的第 5 到 8 轴。



图 3. AMP-19540

## 4. 功能实现

此段详细介绍了如何通过上述已选择的部件来实现控制系统。

### 向量模式

向量模式将两轴联动完成二维的线性及圆弧段。线性及圆弧插补由控制器进行计算，并且通过给定沿路径各处的向量速度，可以勾勒出运动的轮廓（参看命令手册中 VP 命令及 < 和 > 操作数）。下面的命令可以生成图 4 中的图形。

```

VMXY          ; '初始化向量模式
VP 4000,0     ; '给定基座边沿上的第一个直线段
CR 500,270,90 ; '给定基座边沿上的第一个弧线段
VP 4500,2000
CR 500,0,90   ;
VP 0,2500
CR 500,90,90  ;
VP -500,500
CR 500,180,90
VE           ; '结束基座上的序列
BGS

```




图 4. 涂抹过程中通过 WSDK 存储仪记录的 XY 运动轨迹

## 电子齿轮

从喷嘴送出的液体流量的控制与 XY 路径的向量速度直接相关。为了维持每单位行程内液体的涂抹量恒定，我们将泵轴同沿 XY 路径的弧长以电子齿轮方式联系起来。控制器就会自动调整涂抹流量，而我们则不必考虑程序中如何处理沿路径加速与减速时会引起的效果。

## 关键命令：GA，GR

```

GAE = S          ; '定义 S 坐标平面为泵轴 E 的主动轴
GRE = 0.135     ; '设定液泵液体流速与向量速度的齿轮比

```

## 通过多线程对液面进行监测

这台机器需要对液面进行监测以保持对过程进行正确的控制。在另一个线程中，一个循环不停

地扫描着一路监测液面状态的模拟信号。如果液面过低，机器就停止工作，并通过触发报警来警告操作者。

```
XQ #PumpLvl, 1 ;'开始在第二个线程中监测液泵的液面高度

#PumpLvl ; '这个程序通过模拟输入监测液泵液面高度
  IF @AN[1] < 1
    MG "FLUID LEVEL LOW"
    SB1 ;'输出 1 在液面低时启动警报
    AB
  ENDIF
JP #PumpLvl
```

## 终端操作界面

这台机器使用了一个非常简单的文本终端界面。MG 命令用来在显示屏上显示状态信息。IN 命令用来接收用户输入（在本例中为 Z 轴到达基座所要走的距离）。图 5 给出了一段机器运行时屏幕的显示内容。

```
:Homing all axes...
Place new carrier on conveyor
Specify substrate depth (counts)
1000
Moving carrier to start point...
Dispensing...
Dispense has ended
Moving carrier to end point...
Remove finished carrier
Place new carrier on conveyor
```

图 5. 运行时的终端显示

## 程序代码

本段给出了运行机器所需的 Galil 语言程序。程序通过 BP 命令烧到控制器中。由于程序以特殊标号#AUTO 开始，所以当控制器和 PC 机上电时，程序立即执行。程序首先启动液面监测线程 #PumpLvl（与 #AUTO 平行运行）之后使 XY 平面，喷嘴，及传送带回零。当完成这些一次性的初始化之后，#AUTO 进入无限循环 #Loop，在此循环中实现主要的涂抹操作。

程序开始时提示操作者插入新的托盘。当托盘插入传感器（输入 1）探测到托盘进入工作位置后，程序提示用户输入基座的深度。当此数据被输入后，托盘便开始移动直到触发到位传感器（输入 2），此时便开始执行预先定义好的涂抹路径（如图 4）。当路径图形完成后，托盘便移动直到撞到形成末端传感器（输入 3），最后，程序等待用户将托盘卸下，之后开始下一次循环。

---

```
#AUTO
```

```
PmpRatio = 0.1 ;'以 S 平面速度为主轴的液泵齿轮比，液体流量就基于此
```

---

```

CB1 ; '清除液面警报
XQ #PumpLvl, 1 ; '在另一个线程中开始对液泵液面进行监测

'使喷嘴轴 XYZ 轴回零到反向限位并寻找标志脉冲
'使传送带 W 轴回零并寻找标志脉冲
MG "Homing all axes..."

JG*=-10000 ; 'JOG 方式向反向限位运动
HMW ; '传送带回零至零点开关 (CN 确定方向)
BGXYZW ; '开始向限位的运动 (传送带回零)
AMXYZW ; '等待, 直到撞到限位 (传送带寻找标志脉冲)

JGXYZ = 500 ; '向标志脉冲缓慢移动
FIXYZ ; '寻找标志脉冲
BGXYZ ; '开始向标志脉冲的运动
AMXYZ ; '等待直到碰到标志脉冲, 此时位置设为 0.

#Loop
MG "Place new carrier on conveyor"
AI-1 ; '等待托盘入位。 载物盘被放置于传送带起始区域时, 激发输入 1

'要求操作者输入与新基座对应的喷嘴高度规格.
IN "Specify substrate depth (counts)", ZPos

'定位传送带, 以使载物盘与输入 2 的传感器对齐, 指示到达起始位置
MG "Moving carrier to start point..."
JGW = 500
BGW
AI-2
STW
AMW

JS #Dispens
'使载物盘移动到传送带末端并等待直到碰到载物盘末端开关 (输入 3) 之后停止
MG "Moving carrier to end point..."
JGW = 5000
BGW
AI-3
STW

MG "Remove finished carrier"
AI3 ; '等待用户移走到达路径末端已完成加工的载物盘, 使输入 3 回到高电平.
JP #Loop; '跳回之前的传送带程序, 开始对下一个载物盘的操作.
EN

#Dispens

```

```

'position the nozzle to the specified distance from substrate surface
PAZ = ZPos
BGZ
AMZ

GAE = S ;'将泵轴与 S 平面向量序列进行电子齿轮同步
GRE = PmpRatio ;'将之前设定好的数值赋给泵的输液量与向量速度的齿轮比

VMXY      ;'初始化向量模式
VA 250000 ;'设置加速度，减速度，及速度参数
VD 250000
VS 5000
VT 0.8     ;'设置平滑常数

'再此定义基座的涂抹路径
VP 4000,0 ;'给定第一段基座边沿的直线段
CR 500,270,90 ;'给定基座边沿的第一段弧线段
VP 4500,2000
CR 500,0,90
VP 0,2500
CR 500,90,90
VP -500,500
CR 500,180,90
VE ;'终止基座上的序列

MG "Dispensing..."
BGS
AMS

PAZ = 0 ;'当路径全部走完，将停止的喷嘴从基座上抬开
BGZ
AMZ

MG "Dispense has ended"
EN

'*****监测子程序*****

#PumpLvl ; '这个程序通过模拟输入监测泵的液面.
IF @AN[1] < 1
    MG "FLUID LEVEL LOW"
    SB1 ;'输出 1 启动液面过低报警
    AB
ENDIF
JP #PumpLvl
EN

```